@echo off
setlocal
set KEY=HKLM\Software\Microsoft\.NETFramework /reg:64
if /I "%1" EQU "off" goto :off
if /I "%1" EQU "on" goto :on
if /I "%1" EQU "proc" goto :proc
:check
(reg query %KEY%|findstr /ic:"AltJit">nul)2>nul && goto :itson
@echo AltJit is OFF in the registry
goto :exit
:itson
@echo AltJit is ON in the registry
goto :exit
:on
(reg add %KEY% /v AltJit /t REG_SZ /d * /f>nul)2>nul
(reg add %KEY% /v AltJitName /t REG_SZ /d protojit.dll /f>nul)2>nul
goto :check
:off
(reg delete %KEY% /v AltJit /f>nul)2>nul
(reg delete %KEY% /v AltJitName /f>nul)2>nul
goto :check
:proc
tasklist /M protojit.dll
:exit
ENDLOCAL